
ABSTRACT 
In Statistical Programming while working on development of different types of Summary Tables 
and Listings using Proc Report, Statistical Programmers come across a number of challenges as 
far as presentation of the report output is concerned. Some of the challenges faced by the 
Statistical Programmers are: 

1.) Printing certain number of non-missing lines on one page 
2.) Not ending the page with a new parameter starting 
3.) Combining the groups and sub-groups on one page 
4.) Continuation of main group heading in case the splitting of main group and subgroup is 

unavoidable example- adverse events 
5.) Addition of suffix in case of group splitting into different page 

 
INTRODUCTION 
In this paper, we are going to present a Simple Page Break macro, which will help the statistical 
programmer to overcome all the challenges as mentioned above in generation of summary 
reports. The output generated from the Page Break Macro will be a SAS dataset, which in turn 
will be used in the Proc Report Procedure in SAS to generate output as per the Clinical Study 
requirement. This Macro when used in combination with Existing Reporting Macros in the 
Pharmaceutical companies will be quite helpful in cutting down the programming time taken in 
formatting of Tables and Listings. 
 
THE EXAMPLE  
This Page Break Macro is specifically meant for the AE tables where AEDECOD (Adverse Event 
decoded terms) belonging to a particular AESOC (Adverse Event System Organ Class) are being 
spread across multiple pages and it becomes difficult for a reviewer going through these tables 
and listings to keep a track of hierarchical relations between AEDECOD and AESOC. Below is the 
problem we often encountered in Adverse Event Tables: Gastrointestinal disorders is a System 
Organ class under which a number of AEDEOCD terms are present and which are extended on 
the next page once the limit on number of non-missing terms on one page is reached. The Page 
break macro will not only provide an options of having the System Organ class being continued 
on next page alongside its sub-group of AEDECOD terms but it also provide the option of 
printing a certain number of non-missing lines on one Page. 
 



 
 
 

 
Below is the snapshot of the table output by the Proc report procedure, which will be using the 
SAS dataset generated by the Page Break Macro: 
 
 

 



MACRO CODE AND SAMPLE DATASET 
Below is the Snippet of the Page break Macro and following it, the detailed explanation of keyword 
parameter use in the Macro is given. The whole Page Break Macro is provided at the end of the paper. 
 
%macro page_break_ae(input=%nrstr(), main=%nrstr(), sub=%nrstr(), 
                       output=%nrstr(), page_no=, suffix=%nrstr()); 
 
Input = specifies the name of the SAS dataset. 
Main = specifies the name of the variable in the input SAS dataset which act as the main grouping above 
the sub-grouping. In the example mentioned here, this is equal to AEBODSYS or AESOC. 
Sub = specifies the name of the variable in the input SAS dataset which is the sub-group in the hierarchy 
Output = specifies the name of the SAS dataset which will be the output of the Page break Macro.  
Page_no = specifies the number of non-missing observations as per the discretion of user of this Macro 
Suffix = specifies the sentence which can be used in continuation with the main heading which in this 
case is AESOC term and which is continued onto multiple pages.  
 
The input SAS dataset used in the Page Break Macro has to be present in a particular format and is given 
below in the form of the snapshot. In addition, the variables present in it are as follows: 
 

a.) Nam = Variable which represent the hierarchical relation between AESOC and AEDECOD and will 
be present in the Table output 

b.) AEBODSYS = Main Group Term and in this case it is Body System or Organ Class 
c.) AEDECOD = Sub Group Term and in this case it is Adverse Event Dictionary-Derived Term 
d.) PLB = In this case it is Placebo, one of the Treatment Arm 
e.) ACT1 = In this case it is an Active Drug 1 
f.) ACT2 = In this case it is an Active Drug 2 

 



 
 
 
 
 
 
 
 
 



OUTPUT AND PROC REPORT 
Below is a snapshot of the Output SAS dataset, which will be used in Proc Report procedure to generate 
Adverse Event Table. The Extra variables generated in this SAS dataset as compared to the input SAS 
dataset are the key variables which are meant to be used in different statement in Proc Report 
Procedure. 
 
 

 
 
The Proc Report Code mentioned below specifies how the extra variable generated in above mentioned 
output dataset is being used to generate outputs as desired. 
 
proc report data=output; 
column pb_num2 aebodsys nam PLB ACT1 ACT2; 
    define pb_num2  /order noprint; 
    define aebodsys /order order=data noprint; 
    define nam      /display "System Organ Class#~{style~{NBSPACE2}}Preferred Term"; 
    define PLB      /display "Placebo"; 
    define ACT1     /display "Active Drug 1"; 
    define ACT2     /display "Active Drug 2"; 
     
    break after pb_num2 / page; 
    break after aebodsys/skip summarize suppress; 
run; 
 
 

 
 



CONCLUSION 
The Page Break Macro specified in this paper is defined specifically for the Adverse Event Table but it is 
robust enough to be used in generating summarizing reports for other Descriptive Statistical tables for 
Vital Signs, Demographics, and Laboratory etc. In addition, the full code is given below with explanation 
of steps and the user can further modify this Page Break Macro as per the Work requirements.  
 
PAGE BREAK MACRO CODE 
/*****************************PAGE BREAK MACRO****************************/ 
 
 
/***************************KEYWORD PARAMETERS****************************/ 
A.) INPUT    = Name of the input SAS dataset 
B.) MAIN     = Name of the Variable under which Grouping has to be done 
C.) SUB      = Name of the Variable under which Sub-Grouping has to be done  
D.) Output   = Name of output SAS dataset from the processing done by Macro 
E.) PAGE_NO  = Number of Lines to be printed on one RTF page   
F.) SUFFIX   = The Suffix to be used alongside SOC 
/***************************KEYWORD PARAMETERS****************************/ 
 
options mlogic merror serror mprint symbolgen mcompilenote=all; 
%macro page_break_ae(input=%nrstr(adam.dum_tab), main=%nrstr(aebodsys), 

sub=%nrstr(aedecod),output=%nrstr(t14_3_1_4_ae), page_no=, 
suffix=%nrstr(continued)); 

 
*Creating user-defined format from the dataset being given for page break and main 
group continuation; 
proc sort data = &input. out=format_1 nodupkey; 
 by &main.; 
run; 
 
data format_2; 
 length label 8.; 
 set format_1(rename=(&main. = start)); 
 label=_n_; 
 retain fmtname 'Order' type 'C'; 
run; 
 
proc format library=work cntlin=format_2; 
run; 
 
*Setting the page number according to the number of lines to be printed on one 
page; 
data &output._i; 
 set &input.; 

pb_num=ceil(_n_/&page_no.);*PB_NUM is the variable which gives a whole 
number for every observation in the dataset; 
order1=input(put(&main.,$order.),best.);*It gives the order as per the main 
heading in the SOC requirement; 

run; 
 
proc sort data=&output._i; 
 by pb_num &main.; 
run; 
 
data &output._i  

dummy(drop=order2);*Creating Dummy dataset so as main group can be repeated 
on continuing page; 



 set &output._i; 
 by pb_num &main.; 
 if &sub. = '' and last.pb_num then pb_num2 = pb_num + 1;*If main  
group(Parameter) is occuring on the last line of a page then this will take it the 
next page; 
 else pb_num2 = pb_num; 
 
 aa=lag(pb_num2); 

if first.&main. then bb=.; 
 else bb = aa; 
 
 order2=pb_num2; 
 
 if first.&main. and &sub. = ''  then order3 = 1; 
 else order3=2; 
 

output &output._i; 
if &sub. ne '' and bb = . and aa^=pb_num2  then output dummy; 

run; 
 
data dummy2; 
 set dummy; 
 order2=aa + .5; 
 nam=strip(&main.) || " (&suffix.)"; 
run; 
 
proc sql noprint stimer; 
 create table &output.  as  

select * from &output._i 
 outer union corr 
 select * from dummy2(keep=nam order1 order2 order3 &main. pb_num pb_num2 aa bb) 

order by order1, order2 , order3; 
 run; 
quit; 
 
%mend page_break_ae; 
/*****************************PAGE BREAK MACRO****************************/ 


	ABSTRACT
	INTRODUCTION
	THE EXAMPLE
	MACRO CODE AND SAMPLE DATASET
	OUTPUT AND PROC REPORT

	CONCLUSION
	PAGE BREAK MACRO CODE

